Colored 2D Gnuplot map

The NCI method is a valuable tool for visualizing chemical interactions. It uses the properties of the reduced density gradient (s) and the electron density ρ for revealing and characterizing chemical interactions. In order to identify them, s is plotted against ρ multiplied by the sign of the second eigenvalue of the electron density Hessian matrix, $sign(\lambda_2)$. Peaks in the plot identify interaction regions. Then, the points withing these peaks are visualized as 3D s isosurfaces. The identification of chemical interactions is done. To match s *vs* $sign(\lambda_2)\rho$ peaks with the components of the s isosurface, the latter is colored using a red-green-blue color code over some $sign(\lambda_2)\rho$ range. The $sign(\lambda_2)\rho$ interval must be determined from the s *vs* $sign(\lambda_2)\rho$ plot. Sometimes this procedure can be rather tedious, and some interactions lying at very close $sign(\lambda_2)\rho$ values can't be properly resolved.

The following `GNUPLOT` script aims at facilitating the identification of chemical interaction with the NCI method. It maps $sign(\lambda_2)\rho$ over the s *vs* $sign(\lambda_2)\rho$ plot as shown in the figure below. Every peak in the 2D plot is colored in the same color as its corresponding 3D interaction region.



```
# Gnuplot script for mapping NCI color code over NCI diagrams
set terminal postscript landscape enhanced color "Helvetica" 20
set encoding iso_8859_1
set output "ncicolormap.ps"
set key
set ylabel 's' font "Helvetica, 30"
set pm3d map
# Define a color gradient palette used by pm3d
set palette defined (-0.04 "blue",0.00 "green", 0.04 "red")
set format y "%.2f"
set format x "%.2f"
set format cb "%.2f"
set border lw 4
set xtic -0.04,0.01,0.04 nomirror rotate font "Helvetica"
set ytic 0.0,0.5,2.0 nomirror font "Helvetica"
# set the color bar tics
set cbtic -0.04,0.01,0.04 nomirror font "Helvetica"
set xrange [-0.04:0.04]
set yrange [0.0:2.0]
# set the range of values which are colored using the current palette
set cbrange [-0.04:0.04]
plot 'PhenolDimer.dat' u 1:2:1 w p lw 6 palette t ''
```

This is only a quick description of the commands used in this script. For a more detailed explanation please visit the official site of GNUPLOT `http://www.gnuplot.info/`.
Let's analyze the script line by line.

Exporting the graph

First we need to set the commands for exporting the graph. That is the format of the graph (GIF, JPG, PNG, PostScript, etc) and the output device (a external file or the screen). This is done with the `set terminal` and `set output` commands.

```
set terminal postscript landscape enhanced color "Helvetica" 20
set encoding iso_8859_1
set output "ncicolormap.ps"
```

In the example the PostScript terminal is selected. The orientation of the picture is controlled by the `landscape` option. Since the output is colored we need to trigger the `color` option, otherwise a black and white picture is generated. `enhanced` enables text mode features such as subscripts or superscripts. The font and the text size are specified by `"Helevetica" 20`. The character encoding is selected by `set encoding`, `iso_8859_1` in the example. It is equivalent to the PostScript encoding `ISO-Latin-1`. The command `set output` specifies the name of the output file. In this example `ncicolormap.ps`. Notice that we have just added the .ps PostScript extension to the output file name.

Axes and Legends

Now we can start to work on the plot. Fist we enable the use of keys and legends with the `set key` command.

```
set key
```

Now we need to specify we want to map a function over the plot. This is done with the `set pm3d` command. `set pm3d` creates a third axis with the color code map.

```
set pm3d map
```

`set palette` defines the color scale. The scale has to be adjusted to fit each particular case. Since our objective is to match points in the NCI plot with points in the 3D reduced density gradient representation, it is a good strategy to defined the palette bounds as the density value cutoff selected in `NCIPLOT`. These are controlled with the `CUTOFFS` and `CUTPLOT` keywords. In the above example all points with $sign(\lambda_2)\rho$ value of -0.04 are colored in blue (strong attractive interactions). Weak non-covalent interaction at 0.00 in green, and strong repulsive interactions at 0.04 in red.

```
set palette defined (-0.04 "blue",0.00 "green", 0.04 "red")
```

Now we define the interval on each axis. This is done with the commands `set xrange`, `set yrange` and `set cbrange`. The latter refers to the color axis. To obtain a perfect match with NCI isosurfaces, the bounds defined in `set palette` must fit with `set xrange` and `set cbrange` values. Otherwise, the color code will be shifted respect to $sign(\lambda_2)\rho$ values. What is more, for achieving our purpose all the settings on the `x` and `cb` axes must be identical.

```
set xrange [-0.04:0.04]
set yrange [0.0:2.0]
set cbrange [-0.04:0.04]
```

Once the axes are defined, we can start to decorate them. First, we want to show tics along the axes. This is done with the `set xtic xmin,xstep,xmax`, `set ytic ymin,ystep,ymax` and `set cbtic cbmin,cbstep,cbmax` commands. They place tics along x,y and the color axis. The tics are set between `xmin/ymin/cbmin`, `xmax/ymax/cbmax` in steps of length `xstep/ystep/cbstep`.

```
set cbtic -0.04,0.01,0.04 nomirror font "Helvetica"
set xtic -0.04,0.01,0.04 nomirror rotate font "Helvetica"
set ytic 0.0,0.5,2.0 nomirror font "Helvetica"
```

Then we want to define the format to be used for tics. This is done with `set format` In the example floating point format with two decimals has been selected for each axis.

```
set format y "%.2f"
set format x "%.2f"
set format cb "%.2f"
```

To put the finishing touches, we define thes axes labels with the `set xlabel`, `set ylabel` and `set cblabel` commands. They specify the labels on the x, y and color axis respectively. In the `NCIPLOT` diagrams y axis correspond to the reduced density gradient (s) and x axis to $sign\lambda_2\rho$. The color axis (cb) corresponds to the function we want to map over the plot, in this case $sign\lambda_2\rho$. Along with the text, the font and the text size can be also specified.

```
set ylabel 's' font "Helvetica, 30"
```

`set border` controls the borders of the plot.

```
set border lw 4
```

Plotting the data

Finally, we plot the NCIPLOT data saved in the .dat file. By default NCIPLOT generates a file with the extension .dat (PhenolDimer.dat above). It contains the $sign(\lambda_2)\rho$ values in the first column; the reduced density gradient values in the second. The command below will use that file to plot column 1 ($sign(\lambda_2)\rho$) against the data in column 2, and will color the plot using the column 1 values.

```
plot 'PhenolDimer.dat' u 1:2:1 w p lw 6 palette t ''
```